text: # Maintainer: ArchStrike <team@archstrike.org>
  
  buildarch=1
  
  pkgname=spiderfoot
  pkgver=2.11.0
  pkgrel=1
  groups=('archstrike' 'archstrike-recon')
  pkgdesc='The Open Source Footprinting Tool'
  url='http://spiderfoot.net'
  arch=('any')
  license=('GPL2')
  depends=('python2-lxml' 'python2-mako' 'python2-cherrypy' 'python2-m2crypto' 'python2-netaddr' 'python2-requests' 'python2-beautifulsoup4')
  source=("http://downloads.sourceforge.net/project/${pkgname}/${pkgname}-${pkgver}-src.tar.gz")
  sha512sums=('42d86ae035c92a779df9c6fad7baf96cb632594fe5062710e32e7e3330a84d094aca472642151053852247c68ffb676c54adb17a7f40b171064c78a0af790fb0')
  
  prepare() {
    grep -iRl 'python' ${pkgname}-${pkgver} | xargs sed -i 's|python$|python2|'
  }
  
  package() {
    cd ${pkgname}-${pkgver}
    install -dm755 "${pkgdir}/usr/bin"
    install -dm755 "${pkgdir}/usr/share/${pkgname}"
    cp -a --no-preserve=ownership * "${pkgdir}/usr/share/${pkgname}"
  
  cat > "${pkgdir}/usr/bin/${pkgname}" << EOF
  #!/usr/bin/env bash
  cd /usr/share/${pkgname}
  python2 sf.py "\$@"
  EOF
  
  chmod 755 "${pkgdir}/usr/bin/${pkgname}"
  }
  